
Elixir library to stream zip file contents. Works with remote files. Supports Zip64.
Overview
Unzip tries to solve problem of unzipping files from different types of storage (Aws S3, SFTP server, in-memory etc). It separates type of storage from zip implementation. Unzip can stream zip contents from any type of storage which implements Unzip.FileAccess protocol. You can selectively stream files from zip without reading the complete zip. This saves bandwidth and decompression time if you only need few files from the zip. For example, if a zip file contains 100 files and we only want one file then Unzip access only that particular file
Installation
def deps do
  [
    {:unzip, "~> x.x.x"}
  ]
end
Usage
# Unzip.LocalFile implements Unzip.FileAccess
zip_file = Unzip.LocalFile.open("foo/bar.zip")

# `new` reads list of files by reading central directory found at the end of the zip
{:ok, unzip} = Unzip.new(zip_file)

# Alternatively if you have the zip file in memory as binary you can
# directly pass it to `Unzip.new(binary)` to unzip
#
# {:ok, unzip} = Unzip.new(<<binary>>)

# returns list of files along with metadata
file_entries = Unzip.list_entries(unzip)

# returns decompressed file stream
stream = Unzip.file_stream!(unzip, "baz.png")

# if you want to read the whole file as binary
#
# file_content = Enum.into(stream, <<>>, &IO.iodata_to_binary/1)
Supports STORED and DEFLATE compression methods. Supports zip64 specification.
Sample implementations of Unzip.FileAccess protocol
For Aws S3 using ExAws
defmodule Unzip.S3File do
  defstruct [:path, :bucket, :s3_config]
  alias __MODULE__

  def new(path, bucket, s3_config) do
    %S3File{path: path, bucket: bucket, s3_config: s3_config}
  end
end

defimpl Unzip.FileAccess, for: Unzip.S3File do
  alias ExAws.S3

  def size(file) do
    %{headers: headers} = S3.head_object(file.bucket, file.path) |> ExAws.request!(file.s3_config)

    size =
      headers
      |> Enum.find(fn {k, _} -> String.downcase(k) == "content-length" end)
      |> elem(1)
      |> String.to_integer()

    {:ok, size}
  end

  def pread(file, offset, length) do
    {_, chunk} =
      S3.Download.get_chunk(
        %S3.Download{bucket: file.bucket, path: file.path, dest: nil},
        %{start_byte: offset, end_byte: offset + length - 1},
        file.s3_config
      )

    {:ok, chunk}
  end
end


# Using S3File

aws_s3_config = ExAws.Config.new(:s3,
  access_key_id: ["key_id", :instance_role],
  secret_access_key: ["key", :instance_role]
)

file = Unzip.S3File.new("pets.zip", "pics", aws_s3_config)
{:ok, unzip} = Unzip.new(file)
files = Unzip.list_entries(unzip)

Unzip.file_stream!(unzip, "cats/kitty.png")
|> Stream.into(File.stream!("kitty.png"))
|> Stream.run()

For zip file in SFTP server
defmodule Unzip.SftpFile do
  defstruct [:channel_pid, :connection_ref, :handle, :file_path]
  alias __MODULE__

  def new(host, port, sftp_opts, file_path) do
    :ok = :ssh.start()

    {:ok, channel_pid, connection_ref} =
      :ssh_sftp.start_channel(to_charlist(host), port, sftp_opts)

    {:ok, handle} = :ssh_sftp.open(channel_pid, file_path, [:read, :raw, :binary])

    %SftpFile{
      channel_pid: channel_pid,
      connection_ref: connection_ref,
      handle: handle,
      file_path: file_path
    }
  end

  def close(file) do
    :ssh_sftp.close(file.channel_pid, file.handle)
    :ssh_sftp.stop_channel(file.channel_pid)
    :ssh.close(file.connection_ref)
    :ok
  end
end

defimpl Unzip.FileAccess, for: Unzip.SftpFile do
  def size(file) do
    {:ok, file_info} = :ssh_sftp.read_file_info(file.channel_pid, file.file_path)
    {:ok, elem(file_info, 1)}
  end

  def pread(file, offset, length) do
    :ssh_sftp.pread(file.channel_pid, file.handle, offset, length)
  end
end


# Using SftpFile

sftp_opts = [
  user_interaction: false,
  silently_accept_hosts: true,
  rekey_limit: 1_000_000_000_000,
  user: 'user',
  password: 'password'
]

file = Unzip.SftpFile.new('127.0.0.1', 22, sftp_opts, '/home/user/pics.zip')

try do
  {:ok, unzip} = Unzip.new(file)
  files = Unzip.list_entries(unzip)

  Unzip.file_stream!(unzip, "cats/kitty.png")
  |> Stream.into(File.stream!("kitty.png"))
  |> Stream.run()
after
  Unzip.SftpFile.close(file)
end

Module to get files out of a zip. Works with local and remote files
Overview
Unzip tries to solve problem of accessing files from a zip which is
not local (Aws S3, sftp etc). It does this by simply separating file
system and zip implementation. Anything which implements
Unzip.FileAccess can be used to get zip contents. Unzip relies on
the ability to seek and read of the file, This is due to the nature
of zip file.  Files from the zip are read on demand.
Usage
# Unzip.LocalFile implements Unzip.FileAccess
zip_file = Unzip.LocalFile.open("foo/bar.zip")

# `new` reads list of files by reading central directory found at the end of the zip
{:ok, unzip} = Unzip.new(zip_file)

# Alternatively if you have the zip file in memory as binary you can
# directly pass it to `Unzip.new(binary)` to unzip
#
# {:ok, unzip} = Unzip.new(<<binary>>)

# returns list of files along with metadata
file_entries = Unzip.list_entries(unzip)

# returns decompressed file stream
stream = Unzip.file_stream!(unzip, "baz.png")

# if you want to read the whole file as binary
#
# file_content = Enum.into(stream, <<>>, &IO.iodata_to_binary/1)
Supports STORED and DEFLATE compression methods. Does not support zip64 specification yet

        file_stream!(unzip, file_path, opts \\ [])

      


        Returns decompressed file as a stream of iodata. file_path must be the complete file path within the zip. The file is read in chunks, decompressed concurrently in a streaming fashion.

        Returns list of files metadata. This does not make pread call as metadata is already by new/1.

        Reads zip metadata from the passed zip file.

          @type stream_options() :: {:chunk_size, pos_integer()}

          @type t() :: %Unzip{cd_list: map(), zip: struct()}


      


Struct holding zip related metadata, returned by the new/1.
Public fields:
	zip - Zip struct passed to new/1

Remaining fields are private and must not be accessed directly.

          @spec file_stream!(t(), String.t(), [stream_options()]) :: Enumerable.t()


      


Returns decompressed file as a stream of iodata. file_path must be the complete file path within the zip. The file is read in chunks, decompressed concurrently in a streaming fashion.
Options
	chunk_size - Chunks are read from the source of the size specified by chunk_size. This is not the size of the chunk returned by file_stream! since the chunk size varies after decompressing the. Useful when reading from the source is expensive and you want optimize by increasing the chunk size. Defaults to 65_000

          @spec list_entries(t()) :: [Unzip.Entry.t()]


      


Returns list of files metadata. This does not make pread call as metadata is already by new/1.
See Unzip.Entry for metadata fields

          @spec new(Unzip.FileAccess.t()) :: {:ok, t()} | {:error, term()}


      


Reads zip metadata from the passed zip file.
zip must implement Unzip.FileAccess protocol.
Fetches the list of files present in the zip and other metadata by
reading central directory found at the end of the zip.
Returns Unzip struct which contains passed zip struct and zip
metadata.

File metadata returned by Unzip.list_entries/1
	:file_name - (string) File name with complete path. Directory files will have / at the end of their name

	:last_modified_datetime - (NaiveDateTime) last modified date and time of the file. It will be set to nil if datetime is invalid

	:compressed_size - (positive integer) Compressed file size in bytes

	:uncompressed_size - (positive integer) Uncompressed file size in bytes

          @type t() :: %Unzip.Entry{
  compressed_size: pos_integer(),
  file_name: String.t(),
  last_modified_datetime: NaiveDateTime.t() | nil,
  uncompressed_size: pos_integer()
}

Should be implemented for the module which can read from the file system.

        Reads the chunk of data from the file.



    


    
      
        size(file)

      


        Returns the size of the file in bytes.

          @type chunk() :: binary()

          @type reason() :: term()

          @type t() :: FileAccess.t()

          @spec pread(t(), non_neg_integer(), pos_integer()) ::
  {:ok, chunk()} | {:error, reason()}


      


Reads the chunk of data from the file.
It should return {:ok, binary} where binary is the chunk found at offset with length length

          @spec size(t()) :: {:ok, pos_integer()} | {:error, reason()}


      


Returns the size of the file in bytes.

Simple module to access local zip file and example for implementing Unzip.FileAccess.
Note: this is basic implementation, it is callers responsibility to handle resource properly

